
Parcurgând secvenţial şirul de numere, se verifică pentru fiecare număr dacă prin aranjarea 
cifrelor lui se obţine un număr palindrom. Pentru aceasta se construieşte vectorul de apariţii ale 
cifrelor în număr. La un palindrom cu număr par de cifre, fiecare cifră apare de un număr par de ori.  
La unul cu număr impar de cifre, doar o cifră poate să apară de un număr impar de ori şi în rest  
toate de un număr par de ori. Totodată se determină şi numărul de cifre distincte din număr. Dacă 
este palindrom se contorizează şi se păstrează dacă are un număr mai mare de cifre distincte decât 
cel mai bun găsit până atunci sau dacă are acelaşi număr de cifre distincte ca el şi este mai mic. 

O implementare a soluţiei în limbajul Pascal este următoarea:

var f,g:text;
    v:array[0..9] of byte;
    n,i,j,x,c,sf,vip,max:longint;
    impar,cifre,unu:byte;
    cazp:boolean;
begin
     assign(f,'talent.in');reset(f);
     assign(g,'talent.out');rewrite(g);
     readln(f,n);
     max:=0;sf:=0;vip:=0;
     for i:=1 to n do begin
         readln(f,x); c:=x;
         for j:=0 to 9 do v[j]:=0;
         while c>0 do begin
               inc(v[c mod 10]); c:=c div 10
         end;
         impar:=0; cifre:=0; unu:=0;
         for j:= 0 to 9 do begin
             if odd(v[j]) then inc(impar);
             if v[j]=1 then inc(unu);
             if v[j]>0 then inc(cifre)
         end;
         cazp:=(cifre=2) and (v[0]>0) and (v[0] mod 2=0) and (unu=1);
         if (impar<2) and not cazp then begin

inc(sf);
if cifre>max then begin max:=cifre; vip:=x end

                      else if (cifre=max) and (x<vip) then vip:=x
         end
     end;
     writeln(g,sf,' ',vip);
     close(g)
end.

